NAG Library Routine Document

G08AKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G08AKF calculates the exact tail probability for the Mann–Whitney rank sum test statistic for the case where there are ties in the two samples pooled together.

2 Specification

```
SUBROUTINE G08AKF (N1, N2, TAIL, RANKS, U, P, WRK, LWRK, IWRK, IFAIL)

INTEGER             N1, N2, LWRK, IWRK(2*(N1+N2+1)), IFAIL
REAL (KIND=nag_wp) RANKS(N1+N2), U, P, WRK(LWRK)
CHARACTER(1)        TAIL
```

3 Description

G08AKF computes the exact tail probability for the Mann–Whitney U test statistic (calculated by G08AHF and returned through the parameter U) using a method based on an algorithm developed by Neumann (1988), for the case where there are ties in the pooled sample.

The Mann–Whitney U test investigates the difference between two populations defined by the distribution functions $F(x)$ and $G(y)$ respectively. The data consist of two independent samples of size n_1 and n_2, denoted by $x_1, x_2, \ldots, x_{n_1}$ and $y_1, y_2, \ldots, y_{n_2}$, taken from the two populations.

The hypothesis under test, H_0, often called the null hypothesis, is that the two distributions are the same, that is $F(x) = G(x)$, and this is to be tested against an alternative hypothesis H_1 which is

H_1: $F(x) \neq G(y)$; or

H_1: $F(x) < G(y)$, i.e., the x's tend to be greater than the y's; or

H_1: $F(x) > G(y)$, i.e., the x's tend to be less than the y's,

using a two tailed, upper tailed or lower tailed probability respectively. You select the alternative hypothesis by choosing the appropriate tail probability to be computed (see the description of parameter TAIL in Section 5).

Note that when using this test to test for differences in the distributions one is primarily detecting differences in the location of the two distributions. That is to say, if we reject the null hypothesis H_0 in favour of the alternative hypothesis H_1: $F(x) > G(y)$ we have evidence to suggest that the location, of the distribution defined by $F(x)$, is less than the location of the distribution defined by $G(y)$.

G08AKF returns the exact tail probability, p, corresponding to U, depending on the choice of alternative hypothesis, H_1.

The value of p can be used to perform a significance test on the null hypothesis H_0 against the alternative hypothesis H_1. Let α be the size of the significance test (that is α is the probability of rejecting H_0 when H_0 is true). If $p < \alpha$ then the null hypothesis is rejected. Typically α might be 0.05 or 0.01.

4 References

Conover W J (1980) *Practical Nonparametric Statistics* Wiley

Neumann N (1988) Some procedures for calculating the distributions of elementary nonparametric teststatistics *Statistical Software Newsletter* **14(3)** 120–126

Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw–Hill

5 Parameters

1: N1 – INTEGER *Input*

On entry: the number of non-tied pairs, n_1.

Constraint: N1 ≥ 1.

2: N2 – INTEGER *Input*

On entry: the size of the second sample, n_2.

Constraint: N2 ≥ 1.

3: TAIL – CHARACTER(1) *Input*

On entry: indicates the choice of tail probability, and hence the alternative hypothesis.

TAIL = 'T'
A two tailed probability is calculated and the alternative hypothesis is $H_1 : F(x) \neq G(y)$.

TAIL = 'U'
An upper tailed probability is calculated and the alternative hypothesis $H_1 : F(x) < G(y)$, i.e., the x's tend to be greater than the y's.

TAIL = 'L'
A lower tailed probability is calculated and the alternative hypothesis $H_1 : F(x) > G(y)$, i.e., the x's tend to be less than the y's.

Constraint: TAIL = 'T', 'U' or 'L'.

4: RANKS(N1 + N2) – REAL (KIND=nag_wp) array *Input*

On entry: the ranks of the pooled sample. These ranks are output in the array RANKS by G08AHF and should not be altered in any way if you are using the same n_1, n_2 and U as used in G08AHF.

5: U – REAL (KIND=nag_wp) *Input*

On entry: U, the value of the Mann–Whitney rank sum test statistic. This is the statistic returned through the parameter U by G08AHF.

Constraint: U ≥ 0.0.

6: P – REAL (KIND=nag_wp) *Output*

On exit: the tail probability, p, as specified by the parameter TAIL.

7: WRK(LWRK) – REAL (KIND=nag_wp) array *Workspace*

8: LWRK – INTEGER *Input*

On entry: the dimension of the array WRK as declared in the (sub)program from which G08AKF is called.

Constraint: LWRK $\geq n + n(n+1)(n+m) - \dfrac{n(n+1)(2\times n+1)}{3} + 1$, where $n = \min(\text{N1}, \text{N2})$ and $m = \max(\text{N1}, \text{N2})$.

9: IWRK($2 \times (\text{N1} + \text{N2} + 1)$) – INTEGER array *Workspace*

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N1 < 1,
or N2 < 1.

IFAIL $= 2$

On entry, TAIL $\neq$ 'T', 'U' or 'L'.

IFAIL $= 3$

On entry, U < 0.0.

IFAIL $= 4$

On entry, LWRK is too small.

7 Accuracy

The exact tail probability, p, is computed to an accuracy of at least 4 significant figures.

8 Further Comments

The time taken by G08AKF increases with n_1 and n_2 and the product $n_1 n_2$. Note that the amount of workspace required becomes very large for even moderate sizes of n_1 and n_2.

9 Example

This example finds the Mann–Whitney test statistic, using G08AHF for two independent samples of size 16 and 23 respectively. This is used to test the null hypothesis that the distributions of the two populations from which the samples were taken are the same against the alternative hypothesis that the distributions are different. The test statistic, the approximate Normal statistic and the approximate two tail probability are printed. G08AKF is then called to obtain the exact two tailed probability. The exact probability is also printed.

9.1 Program Text

```
    Program g08akfe

!     G08AKF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g08ahf, g08akf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: p, pexact, u, unor
      Integer                          :: ifail, liwrk, lwrk, lwrk1, lwrk2,    &
                                          mn, n1, n2, nsum
      Logical                          :: ties
      Character (1)                    :: tail
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ranks(:), wrk(:), x(:), y(:)
      Integer, Allocatable             :: iwrk(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'G08AKF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in problem size
      Read (nin,*) n1, n2, tail

!     Calculate sizes of various workspaces
      nsum = n1 + n2
      mn = min(n1,n2)

!     Workspace for G08AHF
      lwrk1 = nsum

!     Workspace for G08AKF
      lwrk2 = mn + mn*(mn+1)*nsum - mn*(mn+1)*(2*mn+1)/3 + 1
      liwrk = 2*nsum + 2

      lwrk = max(lwrk1,lwrk2)
      Allocate (x(n1),y(n2),ranks(nsum),wrk(lwrk),iwrk(liwrk))

!     Read in data
      Read (nin,*) x(1:n1)
      Read (nin,*) y(1:n2)

!     Display title
      Write (nout,*) 'Mann-Whitney U test'
      Write (nout,*)

!     Display input data
      Write (nout,99999) 'Sample size of group 1 = ', n1
      Write (nout,99999) 'Sample size of group 2 = ', n2
      Write (nout,*)
      Write (nout,*) 'Data values'
      Write (nout,*)
      Write (nout,99998) '    Group 1  ', x(1:n1)
      Write (nout,*)
      Write (nout,99998) '    Group 2  ', y(1:n2)

!     Perform test
      ifail = 0
      Call g08ahf(n1,x,n2,y,tail,u,unor,p,ties,ranks,wrk,ifail)
```

```
!     Calculate exact probabilities
      If (ties) Then
        ifail = 0
        Call g08akf(n1,n2,tail,ranks,u,pexact,wrk,lwrk,iwrk,ifail)
      End If

!     Display results
      Write (nout,*)
      Write (nout,99997) 'Test statistic     = ', u
      Write (nout,99997) 'Normal statistic   = ', unor
      Write (nout,99997) 'Tail probability   = ', p
      Write (nout,*)
      If (ties) Then
        Write (nout,*)
        Write (nout,*) 'Ranks'
        Write (nout,*)
        Write (nout,99998) '    Group 1  ', ranks(1:n1)
        Write (nout,*)
        Write (nout,99998) '    Group 2  ', ranks((n1+1):nsum)
        Write (nout,*)
        Write (nout,*)
        Write (nout,99997) 'Exact tail probability = ', pexact
      Else
        Write (nout,*) &
          'There are no ties in the pooled sample so G08AKF was not called.'
      End If

99999 Format (1X,A,I5)
99998 Format (1X,A,8F5.1,2(/14X,8F5.1))
99997 Format (1X,A,F10.4)
    End Program g08akfe
```

9.2 Program Data

```
G08AKF Example Program Data
 16  23  'L'                                  :: N1,N2,TAIL
 13.0  6.0 12.0  7.0 12.0  7.0 10.0  7.0
 10.0  7.0 16.0  7.0 10.0  8.0  9.0  8.0     :: End of X
 17.0  6.0 10.0  8.0 15.0  8.0 15.0 10.0
 15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
 13.0 12.0 13.0 12.0 13.0 12.0 12.0          :: End of Y
```

9.3 Program Results

```
 G08AKF Example Program Results

 Mann-Whitney U test

 Sample size of group 1 =    16
 Sample size of group 2 =    23

 Data values

     Group 1   13.0  6.0 12.0  7.0 12.0  7.0 10.0  7.0
               10.0  7.0 16.0  7.0 10.0  8.0  9.0  8.0


     Group 2   17.0  6.0 10.0  8.0 15.0  8.0 15.0 10.0
               15.0 10.0 14.0 10.0 14.0 11.0 14.0 11.0
               13.0 12.0 13.0 12.0 13.0 12.0 12.0

 Test statistic     =    86.0000
 Normal statistic   =    -2.8039
 Tail probability   =     0.0025


 Ranks

     Group 1   29.5  1.5 24.5  5.0 24.5  5.0 16.0  5.0
               16.0  5.0 38.0  5.0 16.0  9.5 12.0  9.5
```

```
     Group 2   39.0  1.5 16.0  9.5 36.0  9.5 36.0 16.0
               36.0 16.0 33.0 16.0 33.0 20.5 33.0 20.5
               29.5 24.5 29.5 24.5 29.5 24.5 24.5

 Exact tail probability =     0.0020
```